

Seminar 4 – Backtracking in Prolog

1. We are given a sequence $a_1 \dots a_n$ composed of distinct integers numbers. We have to display all the subsequences which have a valley aspect. For example, for the list [5, 3, 4, 2, 7, 11, 1, 8, 6], some of the solutions would be [5, 4, 3, 11], [3, 2, 1, 4, 5, 7, 8], [11, 6, 1, 3, 4, 5], etc.
 - A very inefficient solution is to solve the problem using 2 predicates: one to generate all possible subsequences and one to check if a subsequence is a valley or not. This solution is inefficient because there are many possible candidates and most of them are not valleys. For the list above, there are 986328 possible candidates, but „only“ 8828 are correct solutions.
 - If we have at a given point the subset [5, 3, 11] (which is a valley), there are no other elements we could add to the end of this subset, so we should not generate those subsequences at all.
 - In order to have a more efficient solution, we will combine the generation and the verification steps. We will use a candidate list (a collector variable) in which we will keep the solution built until now. We will add elements one-by-one in this list, but only those elements that can lead us to a correct solution.
 - To generate a valley, we need a decreasing sequence of elements, followed by an increasing sequence of elements. When we add an element to the candidate solution we need to know whether we are on the decreasing or increasing side (this will decide if an element can be added or not). This is why we are going to use an extra parameter (let's call it Direction), which will tell us the current direction:
 - o 0 – for the decreasing part
 - o 1 – for the increasing part
 - The candidate list is in fact a collector variable. In general we do not like to use collector variables when the result has to be a list, because we need that addToEnd function, to add the elements to the end of the list. In our case we can avoid implementing the addToEnd function if we add the elements to the beginning of the collector variable. But this means that we will generate the list in the reverse order. For example:
 - o [7]
 - o [6, 7]
 - o [3, 6, 7]
 - o [4, 3, 6, 7]
 - o [5, 4, 3, 6, 7]
 - When we want to add a new element to the candidate list, we have to consider two things: the direction and the first element of the candidate list (which is the last element that was added in the collector variable). This leads to 4 possible cases:
 - o Direction 1 and element to be added less than the first element of the collector variable (ex. 8 to be added in front of [9, 11]) => add the element and keep the Direction 1.

- Direction 1 and element to be added greater than the first element of the collector variable (ex. 8 to be added in front of [7, 11]) => add the element and make the Direction 0.
- Direction 0 and element to be added greater than the first element of the collector variable (ex. 8 to be added in front of [7, 5, 6]) => add the element and keep Direction 0.
- Direction 0 and element to be added less than the first element of the collector variable (ex. 8 to be added in front of [9, 5, 6]) => impossible
- When is the candidate list a solution? To have a valid valley, we need a decreasing and an increasing sequence. First we generate the increasing sequence, so whenever we are at the decreasing sequence part (meaning the direction is 0) we have a solution. But, in the same time, it is possible that we can also extend this solution to generate other solutions.
- How do we generate the elements to be added in the candidate solution? We need a predicate to return us, one-by-one, all the elements from the original list. It is a non-deterministic predicate

$$element(l_1 l_2 \dots l_n) = \begin{cases} l_1 \\ element(l_2 \dots l_n) \end{cases}$$

```
% element(L:list, E: element)
% flow model: (i,o), (i,i)
% L - initial list
% E - an element from the list
element([H|_], H).
element([_|T], E):-
    element(T, E).
```

We are not allowed to add the same element twice in the candidate solution. How can we check this? We can use the *element* predicate for this as well, but with a different flow model (i,i)

Another solution would be to make the *element* predicate return an element and the list without that element. How would we change it to do this?

$$element(l_1 l_2 \dots l_n) = \begin{cases} (l_1, l_2 \dots l_n) \\ (e, l_1 \cup L), \text{ if } (e, L) = element(l_2 \dots l_n) \end{cases}$$

```
%element2(L:List, E: element, R: list)
%flow model (i, o, o)
%L - initial list
%E - an element from the list
%R - the initial list without the element E
element2([H|T], H, T).
element2([H|T], E, [H|Rez]):-
    element2(T, E, Rez).
```

Now we can write the predicate to generate the solutions. It will have the following parameters:

- L – the input list
- C – the collector variable
- D – the current direction
- R – result (only in Prolog).

$generate(L, c_1 \dots c_n, D)$

$$= \begin{cases} c_1 \dots c_n, & \text{if } D = 0 \\ generate(Rest, E \cup c_1 \dots c_n, 0), & \text{if } (E, Rest) = element2(L), E > c_1 \text{ and } D = 0 \\ generate(Rest, E \cup c_1 \dots c_n, 1), & \text{if } (E, Rest) = element2(L), E < c_1 \text{ and } D = 1 \\ generate(Rest, E \cup c_1 \dots c_n, 0), & \text{if } (E, Rest) = element2(L), E > c_1 \text{ and } D = 1 \end{cases}$$

`generate(L:list, C:list, Direction:integer, R:list)`

```
%flow model (i,i,i,o)
% L - initial list
% C - candidate solution
% Direction - 0 for the decreasing part 1 for the increasing part
% R - result

generate(_, Cand, 0, Cand).
generate(L, [H|Cand], 0, R):-
    element2(L, E, Rest),
    E > H,
    generate(Rest, [E,H|Cand], 0, R).
generate(L, [H|Cand], 1, R):-
    element2(L, E, Rest),
    E < H,
    generate(Rest, [E,H|Cand], 1, R).
generate(L, [H|Cand], 1, R):-
    element2(L, E, Rest),
    E > H,
    generate(Rest, [E,H|Cand], 0, R).
```

We need a predicate to call *generate*, let's call it *start*. Since the three recursive clauses require the division of the candidate solution into first element and rest of the list, we cannot start with an empty list. We will need to generate an element (using the *element2* predicate) and to call *generate* having as candidate list the list made of that element.

If we implement and run *start*, we will see that it returns incorrect solutions as well. It will generate lists which have only the decreasing part. In order to avoid this problem, we will actually generate the first two elements of the candidate list, make sure that they are in the correct order (increasing) and only after that will we call *generate*.

$start(L) = generate(Rest, [E_1, E_2], 1), \text{ if } (E_1, R1) = element2(L), (E_2, Rest) = element2(R1), E_1 < E_2$

```

%start(L:list, Rez: list)
%flow model (i,o) (i,i)
%L - initial list
%Rez - one solution
start(L, Rez):-
    element2(L, E1, Rest),
    element2(Rest, E2, Rest2),
    E1 < E2,
    generate(Rest2, [E1,E2], 1, Rez).

```

The *start* predicate generates the solutions one after the other, and for seeing the next solution we have to press ; after every solution. If we want to have a list with all the solutions, we can use the *findall* predicate from Prolog.

$$startAll(l_1 l_2 \dots l_n) = \bigcup start(l_1 l_2 \dots l_n)$$

```

%startAll(L:list, Rez: list)
%flow model (i,o) (i,i)
%L - initial list
%Rez - list with all the solutions
startAll(L, Rez):-
    findall(R, start(L, R), Rez).

```

What if we wanted to have in the solution the elements in the same order in which we have them in the initial list? How should we modify our solution to do this?

When we add an element to the candidate solution, the next elements that will be added in the candidate solution will be placed in front of it. And if we want to keep the original order of element, we need to make sure that the next element to be added will be generated only out of the elements that are in front of the current element in the list.

So if `element2([5, 3, 4, 2, 7, 11, 1, 8, 6])` returns the element 7, we need to make sure that the next element will be generated only from the list [5, 3, 4, 2].

To do this, we will modify the `element2` predicate, to return an element and the list of element before it.

$$element3(l_1 \dots l_n) = \begin{cases} (l_1, \emptyset) \\ (E, l_1 \cup Rest), \text{ if } (E, Rest) = element3(l_2 \dots l_n) \end{cases}$$

```

%element3(L:list, E:element, R:list)
%flow model (i,o,o), (i,i,i)
%L - initial list
%E - an element from the list L
%R - the list of elements until element E
element3([E|_], E, []).
element3([H|T], E, [H|R]):-
    element3(T,E,R).

```

Lisp:

Given a list with sublists and numerical and non-numerical atoms, find the minimum odd number from it.

$$\text{minimum}(l_1 \dots l_n) = \begin{cases} 100, & \text{if } n = 0 \\ \min(l_1, \text{minimum}(l_2 \dots l_n)), & \text{if } l_1 \text{ is an odd number} \\ \min(\text{minimum}(l_1), \text{minimum}(l_2 \dots l_n)), & \text{if } l_1 \text{ is a list} \\ \text{minimum}(l_2 \dots l_n), & \text{otherwise} \end{cases}$$

```
(defun minimum (l)
  (cond
    ((null l) 100)
    ((and (numberp (car l)) (oddp (car l))) (min (car l) (minimum (cdr l))))
    ((listp (car l)) (min (minimum (car l)) (minimum (cdr l))))
    (t (minimum (cdr l))))
  )
)
```